
概 要 Haskellで状態を管理する際は、一般的に代数データ型や型クラスが用いられる
が、データが拡張できないか、動的な性質を持たない。そのため Haskellは、複雑な状
態を扱う問題領域には適していないと考えられてきた。一方で、一般的なオブジェクト
指向言語では、オブジェクトを提供することでこの問題領域で成功を収めている。本論
文では、Haskellの言語仕様を変更することなしに、オブジェクト指向言語から着想を得
たオブジェクトを実現する。本論文で提案するオブジェクトは圏を構成し、合成を用い
て継承を表現できる。また、終了する運命にあるオブジェクトやストリーミングなどに
応用でき、複雑な状態を扱うゲームの実装にも使われている。

1 導入

Haskell[1]は純粋関数型言語であり、型システムが純粋な関数と副作用のある関数を分離するこ
とを要請するため、Haskellで書かれたコードにはバグが入り込みにくい。また、この純粋性のお
かげで並行性と並列性を提供できる稀有な言語でもある [2]。さらに、副作用を含む手続きは専用の
型を持っているだけでなく、自分で手続きを定義し、それらの間の変換を記述できる。このように
Haskellには優れた特長がある一方で、複雑な状態を管理するような問題領域には適していないと
考えられてきた。
一方で、オブジェクト指向プログラミング言語 (以下、OOPL)では、内部状態を隠蔽しインター

フェイスを通して統一的に扱えるオブジェクトを提供している。オブジェクトは継承や複製などに
よりデータの拡張性を持ち、複雑で多様な状態を扱うGUIやゲームなどの領域において成功を収め
ている。
そこで本論文では、Haskellの言語仕様には手を加えずに、オブジェクトを表す手法を提案する。

これまで HaskellにOOPLの機能を取り込むためのさまざまな研究がなされてきた。その主流は、
OOPLで書かれたライブラリに対して外部関数呼び出しを可能にし活用する研究 [3, 4]と伝統的な
オブジェクト指向プログラミング (以下、OOP)を模倣する研究 [5]に分類できる。この論文の目的
はこれらの既存研究とは異なり、拡張性のある状態の管理手法を導入し、OOPに相当する記述力を
Haskellで獲得することであり、OOPLが提供する機能を網羅することは目的とはしない。Haskell

は静的型付きでコンパイル型の言語であるため、比較対象として Java、C++、C♯といった静的型
付きOOPLを念頭におく。
この論文は以下のように構成される。まず、第 2章で状態を隠蔽可能なオブジェクトを導入し、

第 3章でオブジェクトの具体例とインスタンス化の方法を示す。オブジェクトの合成とオブジェク
トの圏の構成について第 4章で説明する。第 5章と第 6章では、それぞれ継承・オーバーライドに
相当するオブジェクトの拡張方法と、応用について議論する。最後に、評価と関連研究、および結
論をそれぞれ第 7章と第 8章で述べる。



2 オブジェクトの定義と利用

この章では、本稿で提案するオブジェクトの定義やその性質について説明する。なお、これから
示すオブジェクトや関連する構造は、公開している objectiveパッケージ 1に格納されている。

2.1 オブジェクトの定義

「オブジェクト」とは、内部状態とメソッドを持ち、メッセージを受け取るとメソッドが起動さ
れ、アクションを返すとともに、その内部状態が変化しうる概念を表す。「アクション」とは、m a

のような型を持ち、プログラムや何らかの動作を表現する値である。オブジェクトが受け取るアク
ションを特に「メッセージ」と呼び、メッセージの型を「インターフェイス」と呼ぶ。「メソッド」
とは、メッセージからアクションへの写像である。オブジェクトは、抽象度の高いメッセージ (アク
ション)を受け取り、抽象度のより低いアクションを返す。
我々は、以下のようにオブジェクトを二種類の型をパラメータとして持つようなデータとして定

義する。この型の定義には、HaskellのRank2Types拡張が必要である 2。

newtype Object f g = Object {

runObject :: forall a. f a -> g (a, Object f g)

}

オブジェクトの同値性は余帰納的に定義される。

定義 1 任意の同値性が判定できる型 aについて g aの同値性が判定できる Functor gと、オブ
ジェクト a :: Object f g、b :: Object f gがあるとする。任意メッセージの f :: f aに対
し、fmap snd (runObject a f)と fmap snd (runObject b f)が同値ならば runObject a fと
runObject b fが同値になるとき、aと bは同値である。

提案したオブジェクトは、ミーリマシンの性質と自然性を併せ持つ。以下順に説明する。

2.2 ミーリマシンとの類似性

前述のように、オブジェクトはメッセージを受け取ると、その結果を返すとともに、自分自身の
状態を変化させうる。したがって、メッセージを受け取り結果を返す、ある種のミーリマシンとし
て捉えられる。以下に Haskellのミーリマシンの表現の一つを示す。このミーリマシンの定義と提
案手法の定義は、類似していることが分かる。

newtype Mealy a b = Mealy {

runMealy :: a -> (b, Mealy a b)

}

しかし、Mealyの定義では、メッセージと結果が一つの型しか持てないため、たとえば「文字列を
返すメソッド」と「数値を返すメソッド」の両方を含むオブジェクトを表現できない。その問題を解
決するには、インターフェイスは結果の型をパラメータとして持つ型であるべきであり、種* -> *

を持つはずである。

2.3 自然性

インターフェイスを M、生成するアクションの型を N、結果の型を aとすれば、メソッドの呼び出
しは以下のような型を持つ関数として表される。

forall a. M a -> N a

1http://hackage.haskell.org/package/objective
2https://ghc.haskell.org/trac/haskell-prime/wiki/Rank2Types



Mと Nが関手ならば、これはちょうど関手間の写像である自然変換に対応する。Haskellでは、
自然変換は以下の Natural型として定義できる。

newtype Natural f g = Natural {

runNatural :: forall a. f a -> g a

}

任意の関手 M、N、型 a、値 m :: M a、任意の関数 f、自然変換 nat :: Natural M Nについて、
以下の等式が成立し、これを自然性と呼ぶ。

runNatural nat (fmap f m) ≡ fmap f (runNatural nat m)

インターフェイスが関手である場合、オブジェクトの自然性を定義できる。任意のオブジェクト
obj :: Object M Nについて以下の性質が成り立つ。

runObject obj (fmap f m) ≡ fmap (f *** id) (runObject obj m)

ただし、(***)は射のペアリングをする演算子で、Haskellの標準ライブラリでは Control.Arrow

モジュールで定義されている。

(***) :: (a -> c) -> (b -> d) -> (a, b) -> (c, d)

f *** g = \(x, y) = (f x, g y)

2.4 変換

提案手法が、ミーリマシンと自然変換の両方の表現力を併せ持つことは、Mealyおよび Natural

を Objectに変換する関数 fromNaturalの存在により、明確に示される。なお、以下の Reqの定義
には、GHCのGADTs拡張を使用している。

fromNatural :: Functor g => Natural f g -> Object f g

fromNatural (Natural t) = liftO t

liftO :: Functor g => (forall x. f x -> g x) -> Object f g

liftO t = Object $ fmap (\x -> (x, liftO t)) . t

data Req a b r where

Req :: a -> Req a b b

fromMealy :: Mealy a b -> Object (Req a b) Identity

fromMealy (Mealy t) = Object $ \(Req a) -> let (b, m) = t a in Identity (b, fromMealy m)

Req a b rは aと同型であり、GADT(Generalized Algebraic Data Types)の制約から結果と
して返される値は必ず b の型を持つ。また、任意の x について x は Identity x と同型だから、
Object (Req a b) Identityは Mealy a bと同型であることが分かる。

3 オブジェクトの具体例

この章では、オブジェクトの具体例を挙げる。インターフェイスとして、メッセージ Printと
Incrementを持つ Counter型を用いる。以下に定義を示す。

data Counter a where

Print :: Counter ()

Increment :: Counter Int



3.1 counterオブジェクトの実装と実行

Counter型のメッセージを受け取るオブジェクト counterを以下のように実装する。counterに
Incrementを送ると内部状態が 1増え、Printを送ると現在の内部状態を出力する。

counter :: Int -> Object Counter IO

counter n = Object (handle n) where

handle :: Int -> Counter a -> IO (a, Object Counter IO)

handle n Increment = return (n, counter (n + 1))

handle n Print = print n >> return ((), counter n)

この実装では、counterと handleが相互再帰しているが、この相互再帰は第 4章で説明する合成
により取り除ける。以下は、GHCi3を使って、runObjectで counterのメソッドを呼び出す例で
ある。

> let obj1 = counter 0

> runObject obj1 Print

0

> (_, obj2) <- runObject obj1 Increment

> (_, obj3) <- runObject obj2 Print

1

3.2 インスタンス化

提案したオブジェクトは、メッセージを受け取ると次のオブジェクトを返す。参照を用いれば、
次々に生成されるオブジェクトを一つのインスタンスとして扱える。ここでは参照としてMVar[7]

を使う方法を示す。Haskellで広く使われている参照型として IORefがあるが、複数のスレッドか
らメソッドを呼び出した際に競合する恐れがある。MVarを用いると、メッセージを送ってから結
果が返るまで他のメッセージをブロックできるため、スレッドセーフなインスタンスを表現できる。
インスタンスに対しメッセージを送信し、メソッドを呼び出す演算子 (.-)を以下のように実装

する。

(.-) :: MVar (Object t IO) -> t a -> IO a

m .- e = do

obj <- takeMVar m

(a, obj’) <- runObject obj e

putMVar m obj’

return a

新しいインスタンスの作成には、newMVarが利用できる。これらの操作により、以下のようにし
て参照に基づいたインスタンス化と、メッセージの送信を表現できる。

do i <- newMVar (counter 0)

i .- Print -- 0 を表示
i .- Increment

i .- Increment

i .- Print -- 2 を表示

4 合成

第 2章で述べた counter の実装では、counter と handle が相互再帰によって実装されていた。
counterと handleを分離できれば、handleはメッセージの解釈に集中できるため、handle の再
利用性も高まる。我々はオブジェクトの合成というアイディアで、この分離を実現する。

3https://www.haskell.org/ghc/



4.1 オブジェクトの合成

p :: Object f gと q :: Object g hが与えられたとき、pが生成するアクション gを qに解
釈させるという操作によって、オブジェクトの合成 p @>>@ q :: Object f hを考えることができ
る。本論文では、この演算を合成と呼ぶ。合成演算子 (@>>@)の定義は、以下のように与えられる。

(@>>@) :: Functor h => Object f g -> Object g h -> Object f h

Object m @>>@ Object n = Object $ fmap joinO . n . m

joinO :: Functor h => ((a, Object f g), Object g h) -> (a, Object f h)

joinO ((x, a), b) = (x, a @>>@ b)

合成の単位元は、以下の echoのような、受け取ったメッセージをそのまま返すようなオブジェ
クトとなる (付録Aを参照のこと)。

echo :: Functor f => Object f f

echo = Object (fmap (\x -> (x, echo)))

4.2 合成の例

(@>>@)と用いて、counterと handleを分離する。具体的には、transformersパッケージ [9]で
提供されているモナド変換子 StateTを用い、counterを StateT Int IOで明示的に状態を扱うよ
う定義し直す。このアイディアの骨子を以下に示す。

counter :: Int -> Object Counter IO

counter n = counterE @>>@ variable n

counterE :: Object Counter (StateT Int IO)

variable :: Int -> Object (StateT Int IO) IO

インターフェイスとしてStateT s mを持つオブジェクトは、メッセージとしてget :: Monad m => StateT s m s

と、put :: Monad m => s -> StateT s m ()を受け取れる。典型的な OOPの言葉では、公開
フィールド sを持つオブジェクトとみなせる。
そして、StateT s mのアクションを返すオブジェクトは、状態 sに依存した動作を表す。どち

らも状態が明示的に型に表れているため、操作を拡張できる一方、カプセル化されていない。両者
を合成することによって、状態 sは隠蔽され、カプセル化が実現される。

4.3 状態を保持するオブジェクト

状態を保持するオブジェクトである variableは、以下のように実装できる。

variable :: Monad m => s -> Object (StateT s m) m

variable s = Object $ \m -> runStateT m s >>=

\(a, s’) -> return (a, variable s’)

4.4 状態を利用するオブジェクト

handle自体は状態を持っていないため、単にアクションの自然変換として記述される。第 2.4節
で示した fromNaturalを用いれば、自然変換は自明にオブジェクトに変換できる。

counterE :: Object Counter (StateT Int IO)

counterE = fromNatural (Natural handle)

handle :: Counter a -> StateT Int IO a

handle Increment = do { n <- get; put (n + 1); return n }

handle Print = get >>= \n -> lift (print n)



4.5 オブジェクトの圏

オブジェクトは変換元と変換先の二つのパラメータを持ち、オブジェクト同士を合成できること
から、圏における射のような性質を持つことが類推される。実際に、アクションを対象、オブジェ
クトを射とすると圏をなす。結合則と単位元の存在は、等式変換により証明できる (付録Aを参照
のこと)。

• アクション f、gに対し、オブジェクト Object f gがある。

• 任意のアクションf、g、hに対し、(@>>@) :: Object f g -> Object g h -> Object f h

なる合成関数が存在する。

• 任意のアクション fに対し、アクションをおうむ返しするオブジェクト echo :: Object f f

が存在する。

• 合成は結合的である

• echoは合成の左単位元かつ右単位元である。

アクションとオブジェクトの圏を考えることにより、典型的なOOPの枠組みを超え、広範に性
質を議論できる。
インターフェイスをM、オブジェクトの起こすアクションを IOとすると、オブジェクトはM

から IOへの個々の射に対応し、オブジェクトの型はドメインとコドメインによってのみ決定され
る。そのため、Java、 C♯、C++のような一般的なクラスベースの静的型付きOOPLにおけるクラ
スと違い、型に実装を対応させるものではない。
「M → IOはN → IOである」(is-a関係)は、N からM への単射が存在、つまりN における

任意のメッセージは、M に属するメッセージとして扱えることを意味する。この単射の存在から、
Hom(M, IO)に属するオブジェクトを、Hom(N, IO)に属するオブジェクトに対応させられること
を示しており、射の合成Hom(N,M)×Hom(M, IO) → Hom(N, IO)の特殊な場合として考えら
れる。

5 オブジェクトの拡張

オブジェクトの圏を用いれば、オブジェクトを拡張する典型的な手法である継承およびオーバー
ライドの仕組みも、単純な式によって表される。ここで、拡張する元のオブジェクトを b : B → IO

とおくと、継承とオーバーライドは、以下のように表現できる。

継承 あるオブジェクトを継承した子は、親が受け取る Bに加え、新たな種類のメッセージAを扱
える。したがって、メッセージの和A+Bを受け取るオブジェクトA+B → IOと捉えられ
る。新たなメソッドAの実装は、親のインターフェイスBと親の文脈 IOを使えるべきであ
る。アクションを組み合わせる方法として直和を用い、オブジェクト f : A → IO+Bとして
定義する。

オーバライド メソッドのオーバーライドを、g : B → IO +Bなるオブジェクトによって表す。何
もオーバーライドしない場合として、自明な射 iB : B → IO +Bが存在する。

上の図式において、射 [id, b] ◦ [f, g]が、オーバーライドを伴って継承されたオブジェクトになる。

5.1 メッセージの和の実装

Haskellにおいて、アクション同士の和は以下のように定義される。

data Sum f g a = InL (f a) | InR (g a)

オブジェクト f、gに対する余ペアリング [f, g]は、以下のように実装できる。

(@||@) :: Functor m => Object f m -> Object g m -> Object (Sum f g) m

a @||@ b = Object $ \r -> case r of

InL f -> fmap (fmap (@||@b)) (runObject a f)

InR g -> fmap (fmap (a@||@)) (runObject b g)

しかし、たとえばメッセージセレクタを並べた (G)ADTや、Sumはモナドではないため、オブ
ジェクトはメッセージを一度きりしか受け付けられない。モナドであれば、メッセージそのものが
合成可能になるため、Smalltalkのカスケーディングのように複数のメッセージを送ることが可能
になる。
かといって、メッセージを定義するたびにモナドのインスタンスも定義すると、コードの重複が

発生するため非効率的である。
任意の* -> *の種を持つデータ型にモナドのインスタンスを対応させることができる手法とし

て、Operationalモナド [10]がある。Operationalモナドの定義は以下の通りである。

data Program t a where

Return :: a -> Program t a

Bind :: t a -> (a -> Program t b) -> Program t b

instance Monad (Program t) where

return = Return

Return a >>= k = k a

Bind t c >>= k = Bind t ((>>= k) . c)

liftP :: t a -> Program t a

liftP t = Bind t Return

Programは、任意のインターフェイスを、モナドとして合成可能なインターフェイスに拡張する。
mがモナドならば、オブジェクト Object t mは、Program tを受け取るように変換できる。

sequential :: Monad m => Object t m -> Object (Program t) m

sequential r = Object $ liftM (fmap sequential) . inv r where

inv obj (Return a) = return (a, obj)

inv obj (Bind e cont) = runObject obj e >>= \(a, obj’) -> inv obj’ (cont a)



Program A、Program Bからの射として liftL = liftP . InL、liftR = liftP . InRが存在
する。また、以下に示す余ペアリングの存在から、Program (Sum A B)は和 Sum A Bと同様の性
質を持つことが分かる。

copair :: (Monad m)

=> Object (Program s) m -> Object (Program t) m

-> Object (Program (Sum s t)) m

copair a0 b0 = sequential (go a0 b0) where

go a b = Object $ \r -> case r of

InL f -> fmap (\a’ -> go a’ b) ‘liftM‘ runObject a (liftP f)

InR g -> fmap (\b’ -> go a b’) ‘liftM‘ runObject b (liftP g)

したがって、オブジェクトの圏においては、必要に応じてメッセージの和 Sum A Bの代わりに
Program (Sum A B)を使っても齟齬は生じない。

5.2 デザインパターン

OOPのデザインパターンを提案手法でどのように実現するかの例を示す。Adapter、Proxy、Dec-

oratorパターンは、既存のインターフェイスを変更し、新たなインターフェイスを持つオブジェク
トを作るためのデザインパターンである。
特にProxyパターンとして考えられる例として、第 2章で定義した counterを拡張する例を紹介

する。Printは、呼ばれた回数をカウントするとともに、元のオブジェクトの Printを呼ぶ。ただ
し、5回呼ばれたら “Limit exceeded” と表示し、それ以降元のオブジェクトの Printを呼ばなく
なる。

wrapper :: Int -> Object Counter (Program (Sum Counter IO))

wrapper n = Object $ \r -> case r of

Print

| n < 5 -> liftL Print >> return ((), wrapper (n + 1))

| otherwise -> liftR (putStrLn "Limit exceeded") >> return ((), wrapper n)

Increment -> liftL Increment >>= \x -> return (x, wrapper n)

counter’ :: Object Counter IO

counter’ = wrapper 0 @>>@ sequential (counter @||@ echo)

一方、Template Methodパターンは、アルゴリズムを抽象化し、具体的な実装をサブクラスとし
て実装するデザインパターンである。本稿で提案したオブジェクトの表現では、Adapter、Proxy、
Decorator、Template Methodに本質的な差異はなく、オブジェクトの合成として統一的に表すこ
とができる。

6 応用

この章では、本稿におけるオブジェクトを定命のオブジェクトおよびストリームへ応用すること
について説明する。

6.1 定命のオブジェクト

オブジェクトが永続せず、途中で終了するというケースは頻繁にある。たとえば、アクションゲー
ムならば、倒した敵はフィールドから消えるなどの仕組みが必要である。典型的な OOP の場合、
終了したと分かったオブジェクトへの参照を削除し、ガベージコレクションを待つという方法が使
われている。しかし、すべての参照を削除する操作は自明ではないうえ、「終了した」とプログラム
上で判定してもメソッドを呼ぶことは可能なため、予期せぬバグやメモリリークを引き起こす可能
性がある。



本稿で提案したオブジェクトの表現を使うと、オブジェクト自身が自発的に終了する場合も扱う
ことができる。Object f Maybeはメッセージを受け取ったとき、結果と次の状態が生成されない
可能性があるオブジェクトを表現する。また、Object f (Either a)は最終結果 aを残して終了
するオブジェクトを意味する。特に、後者は newtypeを用いて異なる型 (Mortalと呼ぶ)にすれば、
モナドのインスタンスにすることができる。本論文では、このような途中で終了する運命にあるオ
ブジェクトを、「定命のオブジェクト」と呼ぶ 4。

newtype Mortal f a = Mortal { unMortal :: Object f (Either a) }

mortal :: (forall x. f x -> Either a (x, Mortal f a)) -> Mortal f a

mortal f = Mortal $ Object (fmap (fmap unMortal) . f)

runMortal :: Mortal f a -> f x -> Either a (x, Mortal f a)

runMortal m = fmap (fmap Mortal) . runObject (unMortal m)

instance Monad (Mortal f) where

return a = mortal $ const $ Left a

m >>= k = mortal $ \f -> case runMortal m f of

Left a -> runMortal (k a) f

Right (x, m’) -> return (x, m’ >>= k)

Mortalモナドにおける return aは、メッセージを一切受け取らず結果 aを残して停止するオブ
ジェクトを表す。一方、m >>= kは、mが終了したとき最終結果を kに渡し、得られた新しいオブ
ジェクトで引き継ぐ。オブジェクトの引き継ぎは、ゲームの実装において、「敵が死亡する際に自爆
する」「プレイヤーが死亡したとき交代する」などの動作の表現に対し柔軟な表現を与える。
Eitherの代わりに、EitherT5を使うと、Mortalをモナド変換子として定義できる。これは次節

で利用する。
参照は自発的に消滅できないため、定命のオブジェクトは、参照を用いたインスタンスではうま

く表現できない点に注意する必要がある。定命のオブジェクトは、リストなどのコンテナに直接格
納するのに適している。以下の announce関数は、あるリストに格納されているすべての Mortalに
対しメソッドを呼び出し、その結果を取得する。

announce :: Monad m => f a -> StateT [Mortal f r] m ([a], [r])

announce f = StateT $ return . h . fmap unzip . partitionEithers

. map (‘runMortal‘ f)

where h (a, (b, c)) = ((a, b), c)

announceは、[Mortal f r]を格納する variableオブジェクトへメッセージとして渡すことが
できる。ゲームにおいて敵キャラクターなどの集まりを管理するとき、すべてに対しメソッドを呼
び出し、消滅したキャラクターを削除するという処理はこの announce一つによって実現される。

6.2 ストリームの表現

オブジェクトは、ストリームの生産者や消費者の役割を果たすこともできる。オブジェクトの合
成を用いて、値を返すメソッドを持つオブジェクトは生産者として、また値を受け取るメソッドを
持つオブジェクトは消費者として扱える。そのため、ストリームとして機能させたい概念に対し、
共通のAPIを提供できる。
オブジェクトが受け取るアクションの型が (->) aのとき、そのオブジェクトは生産者として機

能する。オブジェクト Object ((->) a) mにメッセージ f :: a -> bを渡すと、結果 bが返され

4我々が知る限り mortalに対する一般的な日本語の訳が存在しないため、仏教における「定められた寿命」を意味す
る語を借用する。

5http://hackage.haskell.org/package/either



る。つまり、オブジェクトは fに型 aの値を渡す。たとえば、整数を順番に生成していくオブジェ
クトは以下のように表される。

genNaturals :: Monad m => Int -> Object ((->) Int) m

genNaturals n = Object $ \f -> return (f n, genNaturals (n + 1))

逆に、受け取るアクションが (,) aのときは、消費者となる。オブジェクト Object ((,) a) m

にメッセージ (a, x)を渡すと結果 xがそのまま返るが、オブジェクトは aを利用できる。そして、
生産者と消費者とを結合できる。結合演算子 ($$)の実装を以下に示す。

($$) :: (Monad m) => Object ((->) a) m -> Object ((,) a) m -> m x

a $$ b = do

(x, a’) <- runObject a id

((), b’) <- runObject b (x, ())

a’ $$ b’

(->) aや (,) aを受け取る定命のオブジェクトは、有限のストリームを表現する。定命のオブ
ジェクトでも ($$)は利用可能である。EitherTを返すオブジェクトについて、($$)は以下のよう
な型を持ちうる。

($$) :: (Monad m) => Object ((->) a) (EitherT a m)

-> Object ((,) a) (EitherT a m)

-> EitherT a m Void

EitherT a m Voidは、eitherT return absurdの適用によって m aに変換される。ただし、
Voidは値が存在しないデータ型 6で、absurd :: Void -> aは「Voidの値が存在する」という矛
盾から任意の値の存在を示す関数である。
定命のオブジェクトと不死のオブジェクトを結合したい場合、不死の方を定命の形に変換する必

要がある。EitherTの場合、liftO lift :: Object m (EitherT a m)を後ろに合成することで、
定命の型になる。
以下は、ファイルを一行ずつ読み込み、終端に達したらファイルを閉じて終了する定命オブジェ

クト lineReaderと、文字列を一行ずつ書き出すオブジェクト lineWriterを定義する例である。
mainは両者を結合し、ファイル foo.txtを標準出力に書き出す。

import Control.Monad.Trans.Either

import Control.Monad.Trans

import System.IO

import Data.Void

lineReader :: FilePath -> IO (Object ((->) String) (EitherT () IO))

lineReader path = fmap go $ openFile path ReadMode

where

go h = liftO $ \cont -> lift (hIsEOF h) >>= \r -> if r

then lift (hClose h) >> left ()

else lift $ fmap cont $ hGetLine h

lineWriter :: Object ((,) String) IO

lineWriter = liftO $ \(s, x) -> putStrLn s >> return x

main = do

r <- lineReader "foo.txt"

eitherT return absurd $ r $$ (lineWriter @>>@ liftO lift)

6http://hackage.haskell.org/package/void



7 評価と関連研究

Java、C++、C♯が提供するオブジェクトは、抽象データ型からの派生であり、オブジェクトにシ
ンボルを与えることで、メソッドを起動する。一方、提案するオブジェクトはメッセージ指向に基
づいており、オブジェクトがメッセージを受け取るとメソッドを実行する。メッセージはオブジェ
クトとは独立に定義され、ファーストクラスの値である。

7.1 性能

性能の観点から提案したオブジェクトの実用性を確かめるため、音楽ゲームを作成した。この音
楽ゲームでは、複数の音声データをオブジェクトで表現しており、1秒に数万回の頻度で状態が更
新される。また、1000個以上のオブジェクトを扱うようなゲームの記述にも利用している。現在の
一般的なコンピュータでは、両者とも問題なく動作する。

7.2 関連研究

関連した研究として前述のOOHaskellがある。OOHaskellでは、メソッドの数々を型レベルでパ
ラメータ化されたリストに格納することによってオブジェクトを表現している。しかし、フィール
ドやメソッド名を表すラベルは名前以外の型情報を持っていないため、単純なラベルの集まりとし
て型の付いたインターフェイスを表現できない。本稿で提案したオブジェクトは、名前ではなく手
続きというインターフェイスと、それらの変換に着目した構造を持っているため、型が明確なだけ
でなく、合成などのさまざまな演算とその性質が導き出される。
OOHaskellでは各フィールドの可変な状態を扱うのに参照 (IORef)を使うが、提案手法の表現で

は再帰的な構造を使い、状態をオブジェクトそのものに状態を埋め込んでいることも本質的に異な
る。オブジェクトを扱う際に IOを要求しないため、ゲームのキャラクターなど、抽象的かつ複雑
な状態を持つ概念の記述に適している。

7.3 Extensible effects

Extensible effects[12]が提案するような拡張可能なアクションは、アクション間の変換である提
案手法と親和性を持つ。Extensible effectsでは、作用の種類の型レベルリストによってパラメータ
化されたモナド、Effによって作用を記述している。たとえば、文脈から値を取り出す askと、文
脈に値を与える runReaderは、以下のような型を持っている。

ask :: (Typeable e, Member (Reader e) r) => Eff r e

runReader :: Typeable e => Eff (Reader e :> r) w -> e -> Eff r w

オブジェクトは自然変換を内包しているため、以下のように表すことができる。

runReader’ :: Typeable e => e -> Object (Eff (Reader e :> r)) (Eff r)

Foo、Bar、Bazと、それを解釈する objFoo、objBar、objBazを定義した場合を考える。ただし、
Fooは Bazに依存している。これらはオブジェクトであるため、状態を持つことができる。

runFoo :: Member Baz r => Object (Eff (Foo :> r)) (Eff r)

runBar :: Object (Eff (Bar :> r)) (Eff r)

runBaz :: Object (Eff (Baz :> r)) (Eff r)

オブジェクトの合成により、Foo、Bar、Bazの機能を併せ持ったオブジェクトを構築できる。

runFoo @>>@ runBar @>>@ runBaz

:: Object (Eff (Foo :> Bar :> Baz :> r)) (Eff r)

オブジェクトの圏においては、Effモナドは和と同様の性質を持つ。Extensible effectsの手法を
用いることで、第 5章のようなオブジェクトの拡張をより簡易に実現できる。



7.4 表現問題

表現問題 (expression problem)7の観点から、本論文で導入したオブジェクトを他の手法と比較す
る。表現問題とは、静的型付き言語において、新しいデータの追加と、データに対する新しい操作
の追加が両立するような抽象化が難しいという問題である [11]。つまり、以下の二つの項目を、既
存のコードを再コンパイルすることなく、型の安全性を保ったまま実現することは難しいと考えら
れている。

操作の拡張性 既存のデータ型に対して操作を追加できる。

データの拡張性 既存の型に属する新しいデータを追加できる。

さらに、本論文では新たな評価項目として、「振る舞いの動的性」を定義する。振る舞いの動的性
を持つ場合、内部構造がどのように変化しても、インターフェイスが同一であれば継ぎ目なしに扱
える。
例としてエルフ (Elf)とオーク (Orc)という対象を考える。エルフは、悪の呪文を受けると正気

度が減り、正気度が 0になるとオークとなる。オークは、悪の呪文を受けてもオークのままであり、
内部状態を持たない。この例題において比較項目の具体例を示す。

操作の拡張性 　たとえば「治療を受ける」という操作を追加できるか？

データの拡張性 ドワーフ (Dwarf)という新しい対象を追加できるか？

振る舞いの動的性 エルフはオークに変化できるか？

7.4.1 クラスベースのOOPL

一般的なクラスベースの静的型付き OOPLでは、データの拡張性はあるクラスのサブクラスを
増やすことで実現できる。一方で、操作はクラスの定義に結び付けられているため、操作の拡張は
難しい。エルフが自身を返す代わりに、新たにオークのインスタンスを生成するようなメソッドで
も、同一のインターフェイスを持つため、振る舞いの動的性を持っている。

7.4.2 代数的データ型

Haskellのような関数型言語では、Entityのような直和を定義し、エルフとオークを同一の型に
格納する方法が一般的である。その実装の骨子を示す。

data Entity = Elf Int | Orc

spell :: Entity -> Entity

この実装では、型が同じだからエルフがオークになれるため、振る舞いの動的性を持っている。
また、spell以外の操作関数を追加するのは容易である。しかし、新たなデータ型を追加するには、
Entityや操作関数 spellを変更する必要がある。

7.4.3 型クラス

Haskellでは、データの拡張性を実現するために型クラスが使われることがある。以下に型クラ
スを用いた実装の骨子を示す。

data Elf = Elf Int

data Orc = Orc

class Spell s where

spell :: s -> s

7http://homepages.inf.ed.ac.uk/wadler/papers/expression/expression.txt



instance Spell Elf where

spell = ... -- Orc を書けない

instance Spell Orc where

spell = ...

この方法で定義された場合、存在量化を用いて、Elfと Orcを同一の型にまとめることができる。

data Entity = forall s. Spell s => Entity s

型クラスを用いると、新たなデータ型を追加するのは容易である。操作の追加も、別の型クラス
を実装すればよい。しかし、エルフとオークは別の型であり、spellの型は別の型への変化を表現
できないため、エルフはオークになれず、振る舞いの動的性を持たない。
逆に、以下のように操作をデータ型として定義し、型クラスとして内部状態への操作を表現する

方法もある [8]。

newtype Spell a = Spell { spell :: a -> a }

class Action f where

elf :: Elf -> f Elf

orc :: Orc -> f Orc

こちらもデータと操作の拡張性を持つが、Spellの型が内部状態の型によって決まることから、
エルフがオークになる場合を表現できず、こちらの場合も振る舞いの動的性を持たない。

7.4.4 提案手法

本稿で提案したオブジェクトを用いると、以下のように、呪文を表すメッセージ Spellと、それ
を受け取るオブジェクトである、elfと orcが定義される。

data Spell x where

Spell :: Spell ()

type Entity = Object Spell IO

elf :: Int -> Entity

elf 0 = Object $ \Spell -> return ((), orc)

elf n = Object $ \Spell -> return ((), elf (n-1))

orc :: Entity

orc = Object $ \Spell -> return ((), orc)

spell :: [Entity] -> IO [Entity]

spell = mapM update where

update = fmap snd . flip runObject Spell

オブジェクトは単なる値であるため、データの拡張性を持っている。そのため、dwarfのような
新しいオブジェクトを定義することも容易である。

dwarf :: Int -> Entity

dwarf 0 = Object $ \Spell -> return ((), orc)

dwarf n = Object $ \Spell -> return ((), dwarf (n-1))

また、振る舞いの動的性を持っており、return ((), orc)という式が示すように、エルフやド
ワーフがオークに変化できることが分かる。オブジェクトは、値レベルで振る舞いを組み立てるこ
とが可能なため、以下のような型クラスを定義すれば、操作の拡張性も持たせることができる。



class Elf t where

elf :: Object t IO

instance (Elf s, Elf t) => Elf (Sum s t) where

elf = elf @||@ elf

instance Elf Spell where

elf = …

class Orc t where

orc :: Object t IO

instance (Orc s, Orc t) => Elf (Sum s t) where

orc = orc @||@ orc

instance Orc Spell where

orc = …

elfに対し Cureという操作を追加したい場合、Elf Cureのインスタンスを定義すればよい。す
ると、(Elf s, Elf t) => Elf (Sum s t)のインスタンスから、二種類の操作を受け付けるオブ
ジェクト、elf :: Object (Sum Spell Cure) IOが自動的に得られる。
ここまでで紹介した評価を表にまとめる。

操作の拡張性 データの拡張性 振る舞いの動的性

クラスベースOOP ✓ ✓
代数データ型 ✓ ✓
型クラス ✓ ✓
提案手法 ✓ ✓ ✓

8 結論

本論文では、Haskellで状態を柔軟に取り扱うための新たな手法であるオブジェクトを導入し、そ
の有用性を示すとともに、その応用や代数的性質を議論した。複雑な状態を管理するのに適した
OOPの手法は、データの拡張性と振る舞いの動的性を備えていることを指摘し、また Haskellで
状態を扱う伝統的な手法である代数データ型と型クラスには、それらを同時に有していないことも
示した。本稿で導入したオブジェクトは、表現問題を解決するだけでなく、振る舞いの動的性も備
えている。このオブジェクトは、合成によって振る舞いを拡張または隠蔽していくことが可能であ
り、オブジェクトの部品化や、デザインパターンの表現の基礎となる。本稿で提案したオブジェク
トは圏を構成するので、圏の性質を元に構造を考察できる。さらに、定命のオブジェクトは、既存
のOOPではうまく表現できなかった問題に対して解決方法を与えた。

A 付録

可読性のため、Objectと runObjectをそれぞれ inO、outOと略記する。

A.1 合成の結合則

定理 1 任意のアクションe、f、g、関手h、およびオブジェクト(a :: Object e f)、(b :: Object f g)、
(c :: Object g h)について、結合則 a @>>@ (b @>>@ c) = (a @>>@ b) @>>@ cが成り立つ。

証明 1 等式変換により証明する。

outO (a @>>@ (b @>>@ c))

= {　 (@>>@)の定義 }

fmap joinO . fmap joinO . outO c . outO b . outO a) f

= { fmap fusion }

fmap (joinO . joinO) . outO c . outO b . outO a

= { (joinO . joinO)の展開 }

= fmap (\(((x, ef), fg), gh) -> (x, ef @>>@ (fg @>>@ gh)))

. outO c . outO b . outO a

outO ((a @>>@ b) @>>@ c)

= { (@>>@)の定義 }

fmap joinO . outO c . (fmap joinO . outO b . outO a) f

= { outO . inO = id }

fmap joinO . outO c . fmap joinO . outO b . outO a

= { 自然性 }

fmap joinO . fmap (first joinO) . outO c . outO b . outO a

= { fmap fusion }

fmap (joinO . first joinO) . outO c . outO b . outO a



= { joinO . first joinOの展開 }

= fmap (\(((x, ef), fg), gh) -> (x, (ef @>>@ fg) @>>@ gh))

. outO c . outO b . outO a

= { 余帰納法 }

= fmap (\(((x, ef), fg), gh) -> (x, ef @>>@ (fg @>>@ gh)))

. outO c . outO b . outO a

= { 左辺 }

outO (a @>>@ (b @>>@ c))

A.2 左単位元

定理 2 任意のオブジェクト obj :: Object f gについて、echo @>>@ obj = objである。

証明 2 等式変換により証明する。

outO (echo @>>@ obj)

= { echoの定義 }

fmap (\x -> (x, echo))) @>>@ obj

= { (@>>@)の定義 }

fmap joinO . outO obj . fmap (\x -> (x, echo))

= {　自然性 }

fmap joinO . fmap (first (\x -> (x, echo))) . outO obj

= { fmap fusion }

fmap (joinO . first (\x -> (x, echo))) . outO obj

= { joinOの定義 }

fmap (\(x, m) -> (x, echo @>>@ m)) . outO obj

= { 余帰納法 }

fmap (\(x, m) -> (x, m)) . outO obj

= { fmap id = id }

outO obj

A.3 右単位元

定理 3 任意のオブジェクト obj :: Object f gについて、obj @>>@ echo = objである。

証明 3 等式変換により証明する。

outO (obj @>>@ echo)

= { echoの定義 }

outO (obj @>>@ Object (fmap (\x -> (x, echo))))

= { (@>>@)の定義 }

fmap joinO . fmap (\x -> (x, echo)) . outO obj

= { 自然性 }

fmap (joinO . (\x -> (x, echo))) . outO obj

= { fmap fusion }

fmap (\(x, m) -> (x, m @>>@ echo)) . outO obj

= { 余帰納法 }

fmap (\(x, m) -> (x, m)) . outO obj

= { fmap id = id }

outO obj


